Лабораторный практикум 3

Компьютерная математика
БГУ, ММФ, 1 курс, Математика
(научно-педагог. отделение)
доц. Щеглова Н.Л.
2021-09-28

Метод неопределенных коэффициентов

Цель работы: Освоить метод нахождения неопределенных коэффициентов многочлена, 2 способа. Приобрести навыки написания выражений-однострочников. Ознакомиться с функциональным стилем проектирования.

Выполнение работы

1. Способ 1: приравнивание коэффициентов при одинаковых степенях

Задание 1

Написать функцию, вычисляющую частное и остаток от деления многочлена $f_n(x)$ на многочлен $g_m(x)$, *где n и m - степени многочленов,* $n \geqslant m$.

Использовать метод неопределенных коэффициентов, способ приравнивания коэффициентов при одинаковых степенях многочленов.

Вариант задания указывает преподаватель.

1: f: $3 - 2c + 2c^2 + 2c^3 - 5c^4 - c^5 + c^6 - 4c^7$

g: $3 - 5c - 5c^2 - 4c^3$

2 : " f: $-3 + 5b - 3b^2 + b^3 - 5b^4 + 3b^5 + 2b^6 + 4b^7$

g: $2b - b^2 + 2b^3$

3: f: $-3 - 4z + 2z^3 - 5z^4 + 4z^5 - 4z^6 - 5z^7$

g: $-4 + 2z + 5z^2 + 3z^3$

4: f: $2 + 3x + 5x^2 + x^3 - x^4 + 2x^6 - 5x^7$

g: $1 - 2x + 5x^3$

5: f: $4 + 3a - 2a^2 + 3a^3 - 5a^4 - 5a^5 + 2a^6$

g: $4 + 2a + 2a^2 - a^3$

6: f: $-4c + c^2 + 5c^3 + 5c^4 - 5c^5 + 2c^6 - c^7$

g: $-4 + 2c + 2c^2$

7: f: $-2a - a^2 + a^3 - 4a^4 - 2a^5 + 2a^6 + 5a^7$

g: $4 + 2a + 3a^2 + 4a^3$

8: f: $-5 - y + 4y^2 - 3y^3 - 3y^4 - y^5 + 2y^6 - 4y^7$

g: $5 + 2y - 5y^2 + y^3$

9: f: $5 - 2x + 5x^3 - 3x^4 + x^5 + 3x^6 - x^7$

g: $-5 + 3x + 4x^2 - 2x^3$

10: f: $-1 + 4c + 3c^2 + 5c^3 - c^4 - 3c^5 - c^6 + 3c^7$

g: $-5 + c + 4c^2 + c^3$

Выполнение задания 1

Метод неопределенных коэффициентов в применении к данной задаче доступно изложен, в частности, в школьном учебнике [2].

Определение [2]. Пусть $f_n(x)$ и $g_m(x) \neq 0$ – многочлены из множества R[x]. Разделить в R[x] многочлен $f_n(x)$ на многочлен $g_m(x)$ с остатком – это значит найти такие два многочлена $q(x)$ и $r(x)$ из того же множества, что $f_n(x) = g_m(x)$

$q(x) + r(x)$ и степень многочлена $r(x)$ меньше степени многочлена $g_m(x)$ или $r(x) = 0$. Многочлен $q(x)$ называют неполным частным или частным, а многочлен $r(x)$ – остатком от деления $f_n(x)$ на $g_m(x)$.

Теорема (о делении с остатком) [2]. Каковы бы на были многочлены $f_n(x)$ и $g_m(x) \neq 0$ из R[x], всегда можно и притом единственным образом разделить $f_n(x)$ на $g_m(x)$ с остатком.

Метод неопределенных коэффициентов

Следуя теореме о делении с остатком,
при делении $f_n(x)$ на $g_m(x)$ можно записать $f_n(x) = g_m(x)quo + rem$, где

$quo = a_{n-m}x^{n-m} + a_{n-m-1}\, x^{n-m-1} + \ldots + a_1 x + a_0$

$rem = b_{m-1}\, x^{m-1} + b_{m-2}\, x^{m-2} + \ldots + b_1 x + b_0$

– многочлены степени $n\text{-}m$ и $m\text{-}1$ соответственно, с неизвестными, неопределенными коэффициентами a_i, $i = 0, \ldots, n\text{-}m$ и b_j, $j = 0, \ldots, m\text{-}1$.

Чтобы вычислить неопределенные коэффициенты, нужно получить некоторые условия, связывающие неизвестные величины. В методе неопределенных коэффициентов этими условиями, как правило, является система алгебраических уравнений.

Рассмотрим условие $f_n(x) – (g_m(x)quo + rem)=0$, оно должно выполняться для любого значения переменной x.

Если рассматривать левую часть этого условия как многочлен, то удобно предположить, что справа также стоит многочлен степени n с нулевыми коэффициентами при каждой степени x.

Далее, у многочлена левой части $f_n(x) – (g_m(x)quo + rem)$ группируем коэффициенты при степенях x. Каждый из полученных коэффициентов должен быть равным нулю – получаем систему линейных алгебраических уравнений – и решаем ее относительно неопределенных коэффициентов.

Таким образом, задача нахождения частного и остатка от деления многочлена на многочлен сводится к построению многочленов quo и rem с неопределенными коэффициентами и вычислению этих коэффициентов.

Разобьем сформулированное задание на под задания. Каждое из под заданий будем выполнять *Mathematica* тремя способами:

а) пошаговые вычисления;

б) построение выражения-однострочника;

в) построение функций пользователя.

Задание 1.1

Построить функцию `Poly`, которая по заданной степени, целому неотрицательному числу, строит многочлен от одной переменной с неопределенными коэффициентами.

Проектируемая функция `Poly` должна также предоставить пользователю возможность указания имени переменной и имен неопределенных коэффициентов.

Выполнение задания 1.2

а)-б) Пошаговые вычисления и построение композиции функций.

Следует отметить, что мономы, составляющие многочлен от одной переменной, имеют одну и ту же структуру. Запишем общий вид монома, используя чистую функцию

```
a[#] x^# &
```

Далее мы построим список мономов, и в полученном выражении заменим голову `List` на `Plus`.

Получим список степеней мономов, затем используем структуру повторения

```
Range[0, 7]
{0, 1, 2, 3, 4, 5, 6, 7}
```

```
a[#] x^# & /@ Range[0, 7]
```

$\{a[0], x\,a[1], x^2 a[2], x^3 a[3], x^4 a[4], x^5 a[5], x^6 a[6], x^7 a[7]\}$

```
Plus @@ (a[#] x^# & /@ Range[0, 7])
```

$a[0] + x\,a[1] + x^2 a[2] + x^3 a[3] + x^4 a[4] + x^5 a[5] + x^6 a[6] + x^7 a[7]$

Полученное выражение может быть правой частью глобального правила (`SetDelayed`), которое мы построим для определения функции **`Poly`**.

в) Построение функции пользователя.

Напишем спецификацию функции **`Poly`**.

Функция **`Poly[n,x,a]`** строит многочлен степени **n** от одной переменной **x** с неопределенными коэффициентами *a[i], i = 0,…, n*. Второй и третий аргументы могут не указываться, тогда они имеют заданные по умолчанию значения, **x** и **a** соответственно.

```
Poly[n_Integer?NonNegative, x_Symbol: x, a_Symbol: a] :=
  Plus @@ (a[#] x^# & /@ Range[0, n])
```

Проверим введенное посредством `SetDelayed` (**`:=`**) глобальное правило, его *Mathematica* ассоциировала с символом **`Poly`**

```
?Poly
```

```
Global`Poly
```

```
Poly[n_Integer?NonNegative, x_Symbol : x, a_Symbol : a] :=
 Plus @@ (a[#1] x^#1 &) /@ Range[0, n]
```

Тестируем введенное глобальное правило

```
Poly[7]
```

$a[0] + x\,a[1] + x^2\,a[2] + x^3\,a[3] + x^4\,a[4] + x^5\,a[5] + x^6\,a[6] + x^7\,a[7]$

```
Poly[7, t]
```

$a[0] + t\,a[1] + t^2\,a[2] + t^3\,a[3] + t^4\,a[4] + t^5\,a[5] + t^6\,a[6] + t^7\,a[7]$

```
Poly[7, x, b]
```

$b[0] + x\,b[1] + x^2\,b[2] + x^3\,b[3] + x^4\,b[4] + x^5\,b[5] + x^6\,b[6] + x^7\,b[7]$

```
Poly[0]
```

$a[0]$

Задание 1.2

Записать частное и остаток от деления многочлена $f_n(x)$ на многочлен $g_m(x)$ в виде многочленов *quo* и *rem* с неопределенными коэффициентами.

Выполнение задания 1.2

а)-б) Пошаговые вычисления и построение композиции функций.
Выполним задачу на конкретном примере, по шагам, всякий раз анализируя результат вычислений.
Пусть заданы: $f_n(x)$ – делимое, многочлен степени n,
$g_m(x)$ – делитель, многочлен степени m, $n \geqslant m$.

$f = x^5 - 6x^3 + 2x^2 - 4;$

$g = x^2 - x + 1;$

```
Exponent[#, x] & /@ {f, g}
```

{5, 2}

```
Subtract @@ (Exponent[#, x] & /@ {f, g})
```

3

```
quo = Poly[Subtract @@ (Exponent[#, x] & /@ {f, g})]
```

```
a[0] + x a[1] + x^2 a[2] + x^3 a[3]

rem = Poly[Exponent[g, x] - 1, x, b]
b[0] + x b[1]
```

в) Построение функции пользователя.

Выполните самостоятельно.

Задание 1.3

Для заданных $f_n(x)$ и $g_m(x)$ вычислить неопределенные коэффициенты многочленов *quo* и *rem* такими, чтобы условие $f_n(x)$ - $(g_m(x)quo + rem)=0$, выполнялось для любых значений переменной *x*.

Выполнение задания 1.3

а) Пошаговые вычисления.

```
f - (g quo + rem)
-4 + 2 x^2 - 6 x^3 + x^5 -
  (1 - x + x^2) (a[0] + x a[1] + x^2 a[2] + x^3 a[3]) - b[0] - x b[1]

f - (g quo + rem) // Expand
-4 + 2 x^2 - 6 x^3 + x^5 - a[0] + x a[0] - x^2 a[0] - x a[1] + x^2 a[1] - x^3 a[1] -
 x^2 a[2] + x^3 a[2] - x^4 a[2] - x^3 a[3] + x^4 a[3] - x^5 a[3] - b[0] - x b[1]

f - (g quo + rem) // Collect[#, x] &
-4 - a[0] + x^2 (2 - a[0] + a[1] - a[2]) + x^5 (1 - a[3]) +
 x^3 (-6 - a[1] + a[2] - a[3]) + x^4 (-a[2] + a[3]) - b[0] + x (a[0] - a[1] - b[1])
```

Условие тождественного равенства нулю многочлена *f - (g*quo + rem)* приводит к решению системы линейных алгебраических уравнений (СЛАУ) относительно искомых неопределенных коэффициентов.

СЛАУ составляется следующим образом: группируются коэффициенты многочлена при одинаковых степенях переменной и для каждого коэффициента (множителя при x^i, $i=0,\ldots,n$) записывается условие равенства нулю этого коэффициента. Чтобы многочлен был тождественно равным нулю, должно выполняться каждое записанное условие – получаем систему уравнений.

Составим систему линейных алгебраических уравнений

```
temp1 = CoefficientList[f - (g quo + rem), x]
{-4 - a[0] - b[0], a[0] - a[1] - b[1],
 2 - a[0] + a[1] - a[2], -6 - a[1] + a[2] - a[3], -a[2] + a[3], 1 - a[3]}
```

```
temp2 = Select[temp1, # =!= 0 &]
{-4 - a[0] - b[0], a[0] - a[1] - b[1],
 2 - a[0] + a[1] - a[2], -6 - a[1] + a[2] - a[3], -a[2] + a[3], 1 - a[3]}
```

```
temp3 = (# == 0 & /@ temp2)
{-4 - a[0] - b[0] == 0, a[0] - a[1] - b[1] == 0, 2 - a[0] + a[1] - a[2] == 0,
 -6 - a[1] + a[2] - a[3] == 0, -a[2] + a[3] == 0, 1 - a[3] == 0}
```

и решим ее, предварительно подготовив список *vars*, содержащий искомые неизвестные

```
vars = CoefficientList[#, x] & /@ {quo, rem} // Flatten
{a[0], a[1], a[2], a[3], b[0], b[1]}
```

```
temp4 = Solve[temp3, vars]
{{a[0] → -5, a[1] → -6, a[2] → 1, a[3] → 1, b[0] → 1, b[1] → 1}}
```

Остается подставить вычисленные значения коэффициентов в искомые многочлены

```
{quo, rem} /. temp4
```

$$\{\{-5 - 6x + x^2 + x^3, 1 + x\}\}$$

б) Построение композиции функций.

Оформим выполненную ранее последовательность действий в виде одного выражения, такое выражение называют функцией-однострочником.

```
f - (g quo + rem) // CoefficientList[#, x] & //
      Select[#, # =!= 0 &] & //
     (# == 0 & /@ #) & //
   Solve[#, vars] & //
  ({quo, rem} /. #) & // First
```

$$\{-5 - 6x + x^2 + x^3, 1 + x\}$$

в) Построение функции пользователя.

Следует отметить, что при работе с многочленом, коэффициенты которого содержат параметры, задача выбора тех значений параметров, при которых этот многочлен тождественно равен нулю, встречается довольно часто.

Сформулируем эту задачу отдельно и решим ее, построив функцию пользователя.

Задача 1

Пусть задан многочлен **poly** от одной переменной, коэффициентами **poly** являются выражения, содержащие параметры.

Постройте функцию, которая вычисляет те значения параметров, при которых многочлен `poly` тождественно равен нулю.

Выполнение задачи 1

Построим функцию **PolyCoeffSolve[poly,x,vars]**, которая для заданного многочлена **poly** от переменной **x** вычисляет те значения параметров **vars**, содержащихся в коэффициентах многочлена, при которых **poly** тождественно равен нулю.

```
PolyCoeffSolve[poly_, x_Symbol, vars_List] := poly //
      CoeffficientList[#, x] & //
     Select[#, # =!= 0 &] & //
    (# == 0 & /@ #) & //
   Solve[#, vars] & // First
```

Проверим работу построенной функции

```
PolyCoeffSolve[f - (g quo + rem), x, vars]
{a[0] → -5, a[1] → -6, a[2] → 1, a[3] → 1, b[0] → 1, b[1] → 1}
```

Используем созданную функцию **PolyCoeffSolve** для построения искомой функции **QuoRem.**

```
QuoRem[f_, g_, x_Symbol] :=
 Block[{quo, rem, vars}, quo = Poly[Exponent[f, x] - Exponent[g, x]];
  rem = Poly[Exponent[g, x] - 1, x, b];
  vars = CoefficientList[#, x] & /@ {quo, rem} // Flatten;
  PolyCoeffSolve[f - (g quo + rem), x, vars] // ({quo, rem} /. #) &]
result = QuoRem[f, g, x]
```

$\{-5 - 6x + x^2 + x^3, 1 + x\}$

Убедимся в правильности результата

```
f - (g result[[1]] + result[[2]]) // Simplify
0
```

```
PolynomialQuotientRemainder[f, g, x]
```

$\{-5 - 6x + x^2 + x^3, 1 + x\}$

```
QuoRem[f, g, x] - PolynomialQuotientRemainder[f, g, x]
{0, 0}
```

Построим булево выражение для проверки правильности решения

```
And @@
 (# == 0 & /@ (PolyQuoRem[f, g, x] - PolynomialQuotientRemainder[f, g, x]))
True
```

2. Способ 2: метод частных значений

Задание 2

Представьте заданное произведение линейных множителей в виде многочлена стандартного вида (2.2). Используйте метод неопределенных коэффициентов, в нем – метод частных значений нахождения коэффициентов.
Решите задачу тремя способами: а) пошаговым выполнением, б) построением фукции-однострочника, в) построением стека функций пользователя.

Вариант задания указывает преподаватель.
Варианты заданий

1	$(6+2x)(5+3x)(2+6x)(5+6x)(7+7x)(5+8x)(10+9x)(5+10x)(13+10x)(6+11x)(6+12x)(5+15x)$
2	$(1+2x)(12+4x)(4+5x)(11+5x)(12+7x)(11+8x)(5+11x)^2(6+12x)(15+14x)$
3	$(6+x)(1+2x)(8+2x)(3+3x)(12+3x)(6+5x)(8+10x)(4+11x)(6+12x)(2+14x)(11+14x)(7+15x)$
4	$(1+x)(2+x)^2(-1+2x)(5+2x)(4+3x)(-2+4x)(1+4x)(1+5x)(2+5x)(4+5x)$
5	$(-2+x)^2(-1+x)(9+x)(5+2x)(2+3x)(4+3x)^2(9+3x)$
6	$(-4+x)(2+x)(8+3x)(3+4x)(8+4x)^2(9+4x)(-4+5x)(-3+5x)(3+5x)(9+5x)$
7	$(5+x)(-2+2x)(1+2x)(5+2x)(-2+3x)(-1+4x)(2+4x)(4+4x)(5+4x)(-4+5x)(-2+5x)(5+5x)$
8	$(8+x)^2(11+x)(7+2x)(8+2x)^2(10+2x)(9+3x)(11+3x)(6+4x)(6+5x)(8+5x)$

9	(6 + x) (10 + x) (7 + 2 x)^2 (6 + 3 x) (7 + 3 x) (8 + 3 x) (11 + 3 x) (8 + 5 x) (9 + 5 x) (10 + 5 x) (11 + 5 x)
10	(7 + 2 x) (11 + 2 x) (9 + 3 x) (8 + 4 x) (10 + 4 x) (7 + 5 x) (9 + 5 x)^2 (11 + 5 x)

Выполнение задания 2

Пусть задано произведение линейных множителей с постоянными коэффициентами

$$(2x-5)(x+4)(8-5x)(9-4x)(5+7x) \tag{2.1}$$

Требуется, не перемножая множители, представить (2.1) в виде многочлена от одной переменной x:

$$P_n(x) = a_0 + a_1x + a_2x^2 + \cdots + a_{n-1}x^{n-1} + a_nx^n, a_i \in R, i = 0, \dots, n \tag{2.2}$$

Метод неопределенных коэффициентов для этой задачи состоит в следующем. Выражение (2.1) должно тождественно равняться (2.2), для любого значения переменной должно выполняться

$$(2x-5)(x+4)(8-5x)(9-4x)(5+7x) = a_0 + a_1x + \cdots + a_5x^5 \tag{2.3}$$

Степень многочлена с неопределенными коэффициентами, записанного в правой части (2.3), вычисляется по количеству и структуре множителей заданного выражения (2.1), с учетом их степени и кратности.

В вариантах 1 – 10 задания 2 все множители являются линейными многочленами. Если кратность каждого множителя равна 1, то степень заданного многочлена можно вычислить, используя функцию `Length`. В случае, когда существуют линейные множители кратности выше 2, нужно глубже исследовать структуру заданного произведения.

```
expr = (2 x - 5) (x + 4) (8 - 5 x) (9 - 4 x) (5 + 7 x)
Poly[n_Integer ? NonNegative, x_Symbol: x,
  a_Symbol: a] := Plus @@ (a[#] x^# & /@ Range[0, n])
PolyUnkn = Poly[Length[expr]]
a[0] + x a[1] + x^2 a[2] + x^3 a[3] + x^4 a[4] + x^5 a[5]
```

Теперь, чтобы найти неопределенные коэффициенты многочлена, нужно составить систему уравнений относительно искомых неопределенных коэффициентов $a_i \in R, i = 0, \dots ,5$. Для этого используем метод частных значений.

А именно, выражение (2.3) истинно для любого значения переменной x.

```
expr == PolyUnkn
```

$(8-5x)(9-4x)(4+x)(-5+2x)(5+7x) == a[0]+x\,a[1]+x^2 a[2]+x^3 a[3]+x^4 a[4]+x^5 a[5]$

Чтобы получить условия на искомые коэффициенты, задаем (*n+1*) различных значений переменной *x*, последовательно подставляя эти значения в (2.3). Например,

```
expr == PolyUnkn /. x → 0
```

$-7200 == a[0]$

Получаем (*n+1*) условие для вычисления (*n+1*) неопределенного коэффициента $a_i \in R, i = 0, \ldots, n$.
Если вычислять «вручную», удобно брать значения переменной *x* такие, чтобы выполнять небольшое количество арифметических операций. Так, нули каждого линейного множителя правой части позволяют не вычислять левую часть тождества (2.3).

```
Solve[expr == 0]
```

$\left\{\{x\to -4\},\left\{x\to -\frac{5}{7}\right\},\left\{x\to \frac{8}{5}\right\},\left\{x\to \frac{9}{4}\right\},\left\{x\to \frac{5}{2}\right\}\right\}$

```
expr == PolyUnkn /. Solve[expr == 0]
```

$$\left\{0 == a[0]-4a[1]+16a[2]-64a[3]+256a[4]-1024a[5],\right.$$
$$0 == a[0]-\frac{5a[1]}{7}+\frac{25a[2]}{49}-\frac{125a[3]}{343}+\frac{625a[4]}{2401}-\frac{3125a[5]}{16\,807},$$
$$0 == a[0]+\frac{8a[1]}{5}+\frac{64a[2]}{25}+\frac{512a[3]}{125}+\frac{4096a[4]}{625}+\frac{32\,768a[5]}{3125},$$
$$0 == a[0]+\frac{9a[1]}{4}+\frac{81a[2]}{16}+\frac{729a[3]}{64}+\frac{6561a[4]}{256}+\frac{59\,049a[5]}{1024},$$
$$\left.0 == a[0]+\frac{5a[1]}{2}+\frac{25a[2]}{4}+\frac{125a[3]}{8}+\frac{625a[4]}{16}+\frac{3125a[5]}{32}\right\}$$

Мы же, благодаря *Mathematica*, не обремененные рутинным счетом, можем задавать любые *n+1* значений переменной *x*.
Например,

```
Table[expr == PolyUnkn /. x → i, {i, 0, Length[expr]}]
```

$$\{-7200 == a[0],\ -2700 == a[0]+a[1]+a[2]+a[3]+a[4]+a[5],$$
$$228 == a[0]+2a[1]+4a[2]+8a[3]+16a[4]+32a[5],$$
$$3822 == a[0]+3a[1]+9a[2]+27a[3]+81a[4]+243a[5],$$
$$66\,528 == a[0]+4a[1]+16a[2]+64a[3]+256a[4]+1024a[5],$$
$$336\,600 == a[0]+5a[1]+25a[2]+125a[3]+625a[4]+3125a[5]\}$$

!Еще раз обратите внимание: вообще говоря, степень многочлена не всегда совпадает со значением выражения `Length[expr]`.
Решаем полученную систему линейных алгебраических уравнений, подставляем вычисленные значения неопределенных коэффициентов в искомый многочлен

```
Table[expr == PolyUnkn /. x → i, {i, 0, Length[expr]}];
% // Solve
First@(PolyUnkn /. %)
{{a[0] → -7200, a[1] → -1300,
  a[2] → 9857, a[3] → -3879, a[4] → -458, a[5] → 280}}
```

$-7200 - 1300\,x + 9857\,x^2 - 3879\,x^3 - 458\,x^4 + 280\,x^5$

а) Выполните свой вариант задания пошагово. Проверьте правильность полученного решения, используя встроенную функцию.
б) Сформируйте из пошагового решения функцию-однострочник.
в) Постройте функции пользователя, которые решают поставленную задачу.

3. Разложение многочлена по степеням линейного двучлена

Рассмотрим многочлен $f(x) = b_0 + b_1x + b_2x^2 + \cdots + b_nx^n$.

Разложим многочлен *f(x)* по степеням *(x-x₀)*, для этого представим его в виде

$$f(x) = a_0 + a_1(x - x_0) + a_2(x - x_0)^2 + \cdots + a_n(x - x_0)^n,$$

где $a_0, a_1, \ldots, a_n$ – неопределенные коэффициенты.
Задача сводится к нахождению неизвестных коэффициентов $a_0, a_1, \ldots, a_n$.
Чтобы их вычислить, приравниваем заданный многочлен и искомое разложение. Так как составленное условие должно выполняться для любого x, то можно приравнять коэффициенты при одинаковых степенях переменной x, получить систему уравнений и решить ее относительно искомых неизвестных.

Задание 3.1

Разложить многочлен $f(x) = 1 - 4x - 3x^2 + 2x^3 + x^4$ по степеням двучлена *(x + 1)*, используя метод неопределенных коэффициентов.
Решите задачу тремя способами: пошаговым выполнением, построением фукции-однострочника и построением стека заданных функций.

Выполнение задания 3.1

Полагаем:

$$
\begin{aligned}
&x^4+2x^3-3x^2-4x+1=\\
&=a_4(x+1)^4+a_3(x+1)^3+a_2(x+1)^2+a_1(x+1)+a_0=\\
&=a_4x^4+4a_4x^3+6a_4x^2+4a_4x\ +\ a_4+\\
&+a_3x^3+3a_3x^2+3a_3x+a_3+\\
&+a_2x^2+2a_2x+a_2+a_1x\ +\ a_1+a_0=\\
&=a_4x^4++(4a_4+a_3)x^3+(6a_4+3a_3+a_2)x^2+\\
&+(4a_4+3a_3+2a_2+a_1)x\ +(a_4+a_3+a_2+a_1+a_0)
\end{aligned}
$$

Приравниваем коэффициенты при одинаковых степенях, получаем систему:

$$
\begin{cases}
a_4=1\\
4a_4+a_3=2\\
6a_4+3a_3+a_2=-3\\
4a_4+3a_3+2a_2+a_1=-4\\
a_4+a_3+a_2+a_1+a_0=1
\end{cases}
$$

Решая систему, находим: $a_4=1;\ a_3=-2;\ a_2=-3;\ a_1=4;\ a_0=1.$

а) Решите задачу пошагово. Проверьте правильность полученного решения.

б) Сформируйте функцию-однострочник.

в) Постройте функции пользователя, которые решают поставленную задачу.

Задание 3.2

Разложить многочлен $f(x) = x^4 - 8x^3 + 24x^2 - 50x + 90$ по степеням двучлена $(x - 2)$, используя метод непределенных коэффициентов.

Решите задачу тремя способами: пошаговым выполнением, построением фукции-однострочника и построением стека заданных функций.

Выполнение задания 3.2

Выполните самостоятельно.

Решите задачу пошагово. Проверьте правильность полученного решения.

Сформируйте функцию-однострочник.

Постройте функции пользователя, которые решают поставленную задачу.

Ответ: $f(x) = (x - 2)^4 - 18(x - 2) + 3$.

Литература

1. Голубева Л.Л., Малевич А.Э., Щеглова Н.Л. Компьютерная математика. Символьный пакет *Mathematica*. Лаб. практикум в 2 ч. Ч 1. – Минск: БГУ, 2012. – 235 с.

2. К.О. Ананченко, В.С. Коваленко, П.Т. Воробьев [и др.]. Алгебра и начала анализа: учебное пособие для 10-го класса общеобразовательных школ с углубленным изучением математики – 2-е изд., доработанное – Минск: Народная Асвета, 2000. – 541 с. Тема "Многочлены. Деление с остатком".